

Lo Schaff indicator

Fabio Pacchioni ed Enrico Malverti

Fabio.paccioni@gmail.com

Info@enricomalverti.com

Lo **Schaff Trend Cycle Indicator (STC)** è il prodotto della combinazione di uno stocastico lento con un MACD, combinando così in un singolo indicatore un trend-follower e un oscillatore.

L'indicatore è stato sviluppato principalmente per mercati veloci, in particolare per il forex, sebbene possa essere impiegato in ogni mercato. L'inventore dell'STC, Dough Schaff, ha rilasciato il codice e le formule pubblicamente nel febbraio 2008, e si trovano quindi le implementazioni per tutte le principali piattaforme. L'aspetto interessante è che questo indicatore può essere considerato anticipatore del trend, generando falsi segnali molto raramente.

L'approccio dell'STC Indicator è abbastanza originale nella sua concezione. Mai prima era stato sviluppato un indicatore di trend utilizzando una componente ciclica (di solito si prediligono le MA come base per lo sviluppo di indicatori di tendenza). Data la sua capacità di mantenersi in zona di ipercomprato o di ipervenduto, l'uso di questo indicatore va fatto seguendo la sua signal line nei suoi movimenti, e prendendo profitto quando la signal line stessa colpisce il fondo o il top del grafico.

Come funziona

L'indicatore di ciclo rileva i cicli molto prima di quanto faccia il MACD (serie di EMAs con una signal line, storicamente in ritardo), aggiungendo proprio una componente di ciclo basata su N barre precedenti (stocastico, che però genera molti falsi segnali), moltiplicata per un fattore che va da 0 a 1. La combinazione dei 2 (MACD + Stocastico) genera un indicatore che genera segnali di cambiamento di trend velocemente, fornando al contempo pochi falsi segnali.


Singole componenti:

- MACD (Moving Average Convergence / Divergence). $Macd = EMA(12) - EMA(26)$. Linea Del Segnale = $EMA[MACD](9)$.
- Stocastico lento (%D, media a N periodi di %K) offre una misura della deviazione del prezzo attuale dai livelli normali. Come tutti gli oscillatori, lo stocastico offre indicazioni riguardo allo status di ipercomprato/ipervenduto di uno strumento.
 - Fast Stochastic= $\% K = ((\text{chiusura corrente} - \text{minimo assoluto, cioè il più piccolo, del periodo considerato}) / (\text{massimo assoluto del periodo considerato} - \text{minimo assoluto del periodo considerato})) * 100$
 - % D line = media mobile (% K), si possono utilizzare sia ema, sma, wma, molto flessibile.

L'indicatore Schaff ha come settaggio di default le EMA a 23 e 50 periodi, con una componente ciclica che usa una signal line a 10 periodi.

Schaff Trend Cycle formula:

1. EMAs (1 lunga e 1 corta) della chiusura (tipicamente EMA23 e EMA50).
2. MACD delle EMAs.
3. %K del MACD (10 periodi raccomandato).
4. %D del %K precedente (10 periodi).
5. %K del %D (chiamato PF).
6. %D del PF (PFF)



```

{*****}
Description : Schaff Trend Cycle function
{*****}
[IntrabarOrderGeneration = false]
Inputs: Price( Close), TCLen(10), MA1(23), MA2(50);
Variables: XMac(0), Frac1(0), PF(0), PFF(0), Frac2(0), Factor(.5);

{Calculate a MACD Line}
XMac = MACD(c,MA1,MA2) ;

{1st Stochastic: Calculate Stochastic of a MACD}
Value1 = Lowest(XMac, TCLen);
Value2 = Highest(XMac, TCLen) - Value1;

{%FastK of MACD}
Frac1 = IFF(Value2 > 0, ((XMac - Value1) / Value2) * 100, Frac1[1]);

{Smoothed calculation for %FastD of MACD}
PF = IFF(CurrentBar<=1, Frac1, PF[1] + (Factor * (Frac1 - PF[1])));

{2nd Stochastic: Calculate Stochastic of Smoothed Percent FastD,
PF, above.}
Value3 = Lowest(PF, TCLen); Value4 = Highest(PF, TCLen) - Value3;

{%FastK of PF}
Frac2 = IFF(Value4 > 0, ((PF - Value3) / Value4) * 100, Frac2[1]);

{Smoothed calculation for %FastD of PF}
PFF = IFF(CurrentBar<=1, Frac2, PFF[1] + (Factor * (Frac2 -
PFF[1])));

```

```

{*****Buy and Sell Signals*****}


If PFF crosses above 99 then begin
    buy ("SchaffFC Long") this bar at close;
end
else if PFF crosses below 1 then begin
    sellshort ("SchaffFC Short") this bar at close;
end
else if PFF crosses below 99 then begin
    sell ("SchaffFC Sell") this bar at close;
end
else if PFF crosses above 1 then begin
    buytocover ("SchaffFC Buy") this bar at close;
end;

```


Come tradare lo Schaff trend cycle:

- Cross di STC con i suoi limiti, di solito posti a 25 (30) e 75 (80). Se il valori di STC crossano il limite inferiore, lo strumento è considerato in “sottocomprato” e si effettua un Buy. Viceversa, se STC crossa il limite superiore, lo strumento è in “sottovenduto” si effettua uno Short.
- Ci si può focalizzare anche sulle salite e sulle discese delle curve dell’indicatore. In questo caso quando la curva scende si effettua uno Short, quando sale si effettua un Buy.
- Si può tradare anche in modo diverso, ad esempio osservando gli incroci di STC con una sua media mobile


Esempio di come operare con SCT (FTSEMIB daily)



Drawdown della strategia di base e buon rapporto rischio/rendimento



La robustezza della strategia di base è confermata sul Dax30 sullo stesso time frame (30 minuti)



Da questa buona base si può quindi procedere con l'implementazione di modifiche volte a cercare di contenere i rischi (attraverso una gestione dello stop migliore della strategia di base che gestisca il worst case scenario) e ridurre l'esposizione sul mercato.